ODC 524.11, 524.15, 524.2, 524.315

TAPER-BASED SYSTEM FOR ESTIMATING STEM VOLUMES OF UPLAND OAKS

by Donald E. Hilt

FOREST SERVICE RESEARCH PAPER NE-458
1980

FOREST SERVICE, U.S. DEPARTMENT OF AGRICULTURE
NORTHEASTERN FOREST EXPERIMENT STATION
370 REED ROAD, BROOMALL, PA 19008

The Author

DONALD E. HILT received a bachelor's degree in forestry from Iowa State University in 1969, and a master of science degree in forestry from Oregon State University in 1975. He joined the Northeastern Forest Experiment Station in 1975 as a statistician in the biometrics group at Upper Darby, Pennsylvania. In 1976, he transferred to Delaware, Ohio, where he is a research forester specializing in studies of growth and yield of managed upland oaks.

MANUSCRIPT RECEIVED FOR PUBLICATION 2 JANUARY 1980

Abstract

A taper-based system for estimating stem volumes is developed for Central States upland oaks. Inside bark diameters up the stem are predicted as a function of dbh_{ib}, total height, and powers and relative height. A Fortran IV computer program, OAKVOL, is used to predict cubic and board-foot volumes to any desired merchantable top dib. Volumes of test trees were predicted with acceptable accuracy. Other uses for the taper-based system are discussed.

INTRODUCTION

NUMEROUS VOLUME TABLES for upland oaks have been constructed in past years. While most are satisfactory for the intended use, few provide the flexibility available with a taper-based system for estimating stem volumes. Users often desire predicted volumes to merchantable top diameters that differ from the published tables. A taper-based system allows the user to calculate volumes to any desired top diameter. Volume in any portion of the stem may also be calculated.

Some of the older volume tables constructed by the alignment chart method cannot be computerized because there is no prediction equation. Other volume tables based on regression equations sometimes fail to produce consistent trends between cubic volumes to various top diameters, or between board-foot and cubic-foot volumes. Logical and consistent trends result when a taper-based system is used.

This paper presents a taper-based system for estimating cubic and board-foot volumes of upland oaks. Diameter breast height inside bark (dbh_{ib}) and total tree height are required to estimate stem volumes with the taper-based system. Diameter breast height outside bark (dbh_{ob}), measured in the field, can be converted to dbh_{ib} using appropriate bark-thickness factors for each oak species. Total height may be measured in the field or estimated with a local height-diameter curve.

DATA

Felled-tree measurements were collected on 418 upland oak trees located in Ohio, Kentucky, Missouri, Indiana, and Illinois. Oak species were white oak (*Q. alba* L.), chestnut oak (*Q. prinus* L.), black oak (*Q. veluntina* Lam.), scarlet oak (*Q. coccinea* Muenchh.), and northern red oak (*Q. rubra* L.) Dbh_{ob} ranged from 2.4 to 25.5 inches and total height from 21 to 116 feet. A systematic sample of 334 trees was selected to develop the taper equation. The remaining 84 trees were used to validate the model (test data set one).

Dbh_{ob} was recorded for each standing tree. The trees were then felled and bucked into 6 to 20 sections. Height above ground and diameter inside bark (dib) were measured on each section. Dbh_{ib} was measured at the section point 4.5 feet above ground.

An additional 124 oak trees from West Virginia and Virginia were used as a second test data set. Each tree was bucked into nine sections and the dib and height above ground were recorded for each section. Dbh_{ob} ranged from 6.0 to 21.9 inches and total height from 48 to 104 feet.

ANALYSIS

Taper equation

The taper equation was constructed from tree measurements by using a modification of the taper model proposed by Bruce et al. (1968). The modified form of the equation is

$$y = x^{3/2} + \sum_{i=1}^{n} \sum_{j=0}^{1} \sum_{k=0}^{1} \beta_{ijk} \left(x^{3/2} - x^{\gamma i}\right) D^j H^k \tag{1}$$

where

$y = d^2/D^2$

$x = (H\text{-}h)/(H\text{-}4.5)$

d = dib at measurement point (inches)

$D = dbh_{ib}$ (inches)

H = total tree height (feet)

h = height at measurement point (feet) and the β's and γ's are constants to be determined. The equation satisfies the logical conditions that d equals zero when h equals H, and d equals D when h equals 4.5 feet. I used dbh_{ib} in the equation because bark thickness varies according to oak species. Dbh_{ob} can be converted to dbh_{ib} using appropriate bark-thickness factors for each species.

Note that the β's can be estimated by linear model procedures for given values of the γ's. I used a simplified version of equation (1),

$$y = x^{3/2} + \sum_{i=1}^{n} \beta_i (x^{3/2} - x^{\gamma i}) \qquad (2)$$

to determine reasonable values for the γ's. Different values of the γ's were substituted into (2) by trial and error, and the resulting equation was fitted to the data with stepwise regression procedures (maximum R^2 improvement method) after each substitution. The values $\gamma_1 = 3$ and $\gamma_2 = 30$ suggested by Bruce proved to be satisfactory. Given these values for the γ's, stepwise regression procedures were then used to estimate the β's in equation (1). The resulting model was

$$\begin{aligned}\hat{y} = x^{3/2} &+ 3.04392(10^{-3})(x^{3/2} - x^3) H \\ &+ 2.26030(10^{-4})(x^{3/2} - x^3) DH \\ &+ 3.21012(10^{-3})(x^{3/2} - x^{30}) D \\ &- 2.12000(10^{-4})(x^{3/2} - x^{30}) DH\end{aligned} \qquad (3)$$

Inside bark diameters on the stem can be estimated from (3) as

$$\hat{d} = D\sqrt{\hat{y}} \qquad (4)$$

Equation (4) represents the average stem profile for trees of specified D and H. The calculated R^2 for the taper model based on residuals of the form $(\widehat{d - d})$ was 0.93. Typical curves shown in Figure 1 illustrate the reasonable taper curves, including butt flare, generated by equations (3) and (4) for upland oaks.

Cubic volumes

A form factor $(\hat{F})$ representing the ratio of cubic volume inside bark from tip to a given height (h^*) on the stem, to a cylinder of diameter D and height (H - 4.5), can be computed by integrating (3):

$$\begin{aligned}\hat{F} &= \int_0^z \hat{y}\, dx \\ &= 0.4 z^{5/2} + 1.21752 (10^{-3}) (z^{5/2}) H \\ &\quad - 7.6098 (10^{-4}) (z^4) H \\ &\quad + 9.041 (10^{-5}) (z^{5/2}) DH \\ &\quad - 5.651 (10^{-5}) (z^4) DH \\ &\quad + 1.28405 (10^{-3}) (z^{5/2}) D \\ &\quad - 1.0355 (10^{-4}) (z^{31}) D \\ &\quad - 8.48 (10^{-5}) (z^{5/2}) DH \\ &\quad + 6.84 (10^{-6}) (z^{31}) DH\end{aligned} \qquad (5)$$

where

$$z = \frac{H-h^*}{H-4.5}$$

Predicted cubic volume inside bark $(\hat{V})$ from tip to h^* is then calculated as

$$\hat{V} = .00545415\, D^2 (H-4.5) \hat{F}. \qquad (6)$$

Cubic volume of the stem from tip to a 1-foot stump (TOTVOL) was predicted with equations (5) and (6) using $h^* = 1.0$. Merchantable cubic volumes from a 1.0-foot stump to merchantable top dib were predicted by setting h^* equal to the height to merchantable top dib, then subtracting the resulting volume from TOTVOL. A stump volume equal to the volume of a cylinder 1 foot long and diameter equal to the predicted dib at 1 foot was added to TOTVOL to arrive at the predicted total stem cubic volume (CVTS).

Figure 1.—Stem profiles as estimated by equations (3) and (4). Constant dbh_{ib}/dbh_{ob} ratio = 0.91 used.



"Actual" cubic volumes of trees were calculated by summing the volumes of measured sections. The volume of each section was computed as a frustrum of a cone. Predicted volumes were then compared to the actual volumes for CVTS and cubic volume to a 4-inch top (CV4) (Table 1). The measured dbh_{ib} was used to predict volumes for the mode data set. Dbh_{ib} was calculated as 0.91 x dbh_{ob} for the two test data sets to simulate actual use of the equations because bark thickness is not usually measured in the field. The average percentage deviations and differences between total volumes were considered acceptable for the model data set, especially since the computation of actual volumes is certainly not exact. Excellent results were also obtained for the first test data set. The standard deviation of the precentage differences was approximately 10 percent in most cases. This value indicates that for a given tree, the CVTS or CV4 can be estimated with a ± 10 percent error. Total volume for all trees is estimated with more accuracy.

Predicted volumes were consistently low for the second test data set. Part of the reason for low estimates may be due to the constant dbh_{ib}/dbh_{ob} ratio of 0.91 that

Table 1.—Comparison of predicted and "actual" volumes

Data set[a]	No. of trees[b]	Average percentage differences[c]	Standard deviation of percentage differences	Total "actual" volume	Total predicted volume	Percentage difference, totals
Model:						
CVTS	334	1.1	9.8	8861	8811	-0.6
CV4	187	1.8	10.6	7051	7005	-0.7
BF8	187	1.7	19.1	35816	34109	-4.8
Test 1:						
CVTS	84	1.8	10.0	2310	2309	0
CV4	49	2.2	9.9	1883	1881	-0.1
BF8	49	0.7	19.1	9577	9274	-3.2
Test 2:						
CVTS	124	-6.9	8.5	3471	3205	-7.7
CV4	59	-5.6	9.0	2643	2452	-7.2
BF8	59	-9.4	15.5	13916	12342	-11.3

[a]CVTS = cubic volume total stem, ib; CV4 = cubic volume to 4-inch top, ib; BF8 = International 1/4-inch rule board-foot volume to 8-inch top.

[b]CVTS analyzed for all trees; CV4 and BF8 for trees 11.6 inches dbh_{ob} and larger.

[c]Percentage difference = [(predicted volume — actual volume)/actual volume] x 100.

was applied. The main reason was because the stem form of the trees from West Virginia and Virginia was much better than the stem form of the trees used to build the taper equation. The average of the stem form factors predicted with equation (5) using h* = 1.0 was 0.472, while the actual stem form averaged 0.509. If the sample of 124 trees provide a good representation of all trees found in West Virginia and Virginia, then the taper equation constructed in this paper for Central States upland oaks would not be applicable and the construction of an appropriate taper system for this geographical region is recommended. If the sample of well-formed trees occurred by chance, as the case may be occasionally for Central States upland oaks, then the predicted volumes can be adjusted by using methods proposed by Gevorkiantz (1955).

Cubic volumes to various merchantable top diameters are shown in Figure 2 for a tree 80 feet tall. The volumes generated from equations (5) and (6) form smooth and logical progressions with increasing dbh_{ob}. Trends were similar for other height classes.

Board-foot volumes

International 1/4-inch rule board-foot volumes were calculated for all trees 11.6 inches dbh_{ob}. and larger. Total board-foot volume (BF8) in the tree was determined by summing the board-foot volumes of all logs in the tree to an 8-inch top. Allowance was made for a 1.0-foot stump and 0.3 feet of trim for each 16-foot log. Fractional 16-foot log lengths at the top of the tree were calculated by 4-foot sections, including fractional length on the remaining 4-foot section. Predicted board-foot volumes were computed using scaling diameters estimated with equations (3) and (4). Actual board-

Figure 2.—Cubic volumes to 8-, 6-, 4-inch tops, and total stem for 80-foot total height tree.



foot volumes were computed using measured or interpolated scaling diameters.

A comparison of actual and predicted board-foot volumes is shown in Table 1. Average percentage deviations and differences between total values were acceptable for the model data set and the first test data set. The standard deviation of the percentage differences, 19.1 in both cases, is larger than it is for cubic volumes. This larger value should be expected because estimation of board-foot volumes in trees requires the estimation of several scaling diameters up the stem. Considering the variability that occurs in many upland oak stems, the results are excellent.

Estimates for the second test data set are low for board-foot volumes, just as they were for cubic volumes. Predicted board-foot volumes may also be adjusted according to methods proposed by Gevorkiantz (1955).

Board-foot/cubic-foot ratios were plotted for various heights and diameters (Figure 3). The ratios follow smooth and logical trends.

Program OAKVOL

The taper-based volume equations presented in this paper are too complex to be solved easily with a desk calculator. A Fortran IV computer program titled OAKVOL (Appendix) was written to provide the flexibility desired when using the taper-based system. The four subroutines provide the necessary programing logic for applying the taper-based system to a variety of uses. Explanation of each subroutine may be found in the program listing. The main program was written to allow the user to specify any number of fixed merchantable top diameters for both cubic and board-foot volumes. Modification of the main program, and possibly the subroutines, is necessary depending on the intended use.

Figure 3.—International 1/4-inch board-foot/total stem cubic volume ratios.



Table 2.—Sample total cubic-foot volume table [a]

Diameter breast high, ob (inches)	Total height (*feet*)										
	20	30	40	50	60	70	80	90	100	110	120
2	.236	.290	.36	—	—	—	—	—	—	—	—
3	.559	.662	.82	.99	—	—	—	—	—	—	—
4	—	1.194	1.46	1.76	2.1	—	—	—	—	—	—
5	—	1.893	2.30	2.75	3.2	3.7	—	—	—	—	—
6	—	—	3.3	4.0	4.6	5.3	6.0	6.7	—	—	—
7	—	—	4.6	5.4	6.3	7.3	8.2	9.2	—	—	—
8	—	—	6.0	7.1	8.3	9.5	10.7	11.9	—	—	—
9	—	—	7.6	9.0	10.5	12.0	13.5	15.0	16.6	—	—
10	—	—	9.5	11.2	12.9	14.8	16.6	18.5	20.4	—	—
11	—	—	—	13.6	15.7	17.9	20.1	22.3	24.5	26.7	—
12	—	—	—	16.2	18.7	21.2	23.8	26.4	29.1	31.7	—
13	—	—	—	19.0	21.9	24.9	27.9	30.9	33.9	36.9	—
14	—	—	—	22.1	25.5	28.9	32.3	35.7	39.2	42.6	—
15	—	—	—	25.5	29.2	33.1	37.0	40.9	44.8	48.6	—
16	—	—	—	—	33.3	37.6	42.0	46.3	50.7	55.0	59.3
17	—	—	—	—	37.6	42.4	47.3	52.1	56.9	61.7	66.5
18	—	—	—	—	42.2	47.5	52.9	58.2	63.5	68.8	74.0
19	—	—	—	—	47.1	52.9	58.8	64.6	70.5	76.2	81.8
20	—	—	—	—	52.2	58.6	65.0	71.4	78.4	83.9	90.0
21	—	—	—	—	—	64.6	71.5	78.4	85.2	91.9	98.5
22	—	—	—	—	—	70.8	78.3	85.8	93.1	100.3	107.3
23	—	—	—	—	—	77.3	85.4	93.4	101.3	108.9	116.4
24	—	—	—	—	—	84.1	92.8	101.3	109.7	117.9	125.8
25	—	—	—	—	—	91.2	100.4	109.6	118.5	127.1	135.5

[a]Cubic volume of entire stem, less bark. Includes 1.0-foot stump volume.
Constant dbh_{ib}/dbh_{ob} ratio = 0.91 used.

Table 3.—Sample board-foot volume table [a]

Diameter breast high,ob (inches)	Total height (feet)							
	50	60	70	80	90	100	110	120
12	53	65	77	90	101	112	124	—
13	69	85	102	118	134	150	166	—
14	85	105	127	147	168	188	208	—
15	102	126	153	177	202	227	251	—
16	—	149	180	208	238	267	295	323
17	—	173	208	241	276	308	341	373
18	—	198	237	276	314	351	388	424
19	—	224	266	312	354	396	437	477
20	—	251	298	349	395	443	488	532
21	—	—	332	388	438	491	539	588
22	—	—	367	428	484	541	593	647
23	—	—	404	469	531	592	649	706
24	—	—	442	511	579	644	707	767
25	—	—	482	555	630	698	766	829

[a]International 1/4-inch rule board-foot volumes to 8-inch top dib. Allowance made for 1.0-foot stump, 0.3 feet trim for each 16-foot log, and fractional length on upper log. Constant dbh_{ib}/dbh_{ob} ratio = 0.91 used.

Even though a large number of computations are necessary to calculate volumes using OAKVOL, the cost is very reasonable. In a sample run, CVTS was calculated for 33,600 trees and BF8 for 11,628 trees in 38 seconds on an AMDAHL 470V/6 computer. Computing could be simplified by fitting regression equations to explain relationships such as those shown in Figure 2 and 3. However, most users have access to high-speed computers, and should prefer the flexibility available with OAKVOL.

Sample volume tables generated with OAKVOL for CVTS and BF8 are shown in Table 2 and 3. A constant dbh_{ib}/dbh_{ob} ratio of 0.91 was used to determine dbh_{ib}. In practice, a variable ratio according to the oak species should be used.

DISCUSSION

The inherent flexibility of a taper-based system allows compatible cubic and board-foot volumes to be calculated to any desired top diameter. Modification of OAKVOL will allow users to apply the system to a variety of other uses. Percentage distributions of volumes within trees, often desired for biomass and whole-tree chipping studies, can be conveniently computed with a taper-based system. Stem surface areas, ib, can also be calculated. The easiest method for computing surface areas would be to segment the "stem" into a finite number of sections (100 should be adequate), then calculate the surface area for each section with the formula for a truncated cone.

Volume tables based on the number of logs to variable top diameters, often desired for timber cruising, can also be constructed using the taper-based system. Field measurements would require dbh_{ob}, total height, and number of 16-foot logs to the variable merchantable top dib. Total height could be estimated using a local height-diameter relationship. After converting the number of logs to a merchantable height, volumes could be computed using a modified version of OAKVOL.

The taper-based system developed in this paper should be useful to many foresters working in the upland oak timber type. High-speed computers permit inexpensive applications of the system.

LITERATURE CITED

Bruce, David, Robert O. Curtis, and Caryanne Vancoevering.
1968. **Development of a system of taper and volume tables for red alder.** For Sci. 14:339-350.

Gevorkiantz, S.R.
1955. **Composite volume tables for timber and their application in the Lake States.** U.S. Dept. Agric. For. Serv. Tech. Bull. 1104. 51 p.

APPENDIX

```
CXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXC
C                                                                              C
C     PROGRAM OAKVOL                              WRITTEN BY D.HILT, 1979      C
C                                                                              C
C                                                                              C
C     CUBIC FOOT AND BOARD FOOT (INTERNATIONAL 1/4-INCH RULE) VOLUMES          C
C     CAN BE CALCULATED TO VARIOUS MERCHANTABLE INSIDE BARK DIAMETERS          C
C     USING THE SUBROUTINES PRESENTED IN THIS PROGRAM. THE MAIN PROGRAM        C
C     MUST BE MODIFIED BY EACH INDIVIDUAL USER TO FACILITATE THE               C
C     DESIRED INPUT AND OUTPUT. THE PROGRAM PRESENTED HERE ALLOWS A            C
C     VARIABLE NUMBER OF MERCHANTABLE INSIDE BARK DIAMETERS TO BE              C
C     SPECIFIED.                                                               C
C                                                                              C
C     TREES MUST BE LARGER THAN 1.9 INCHES DBHOB AND TALLER THAN 20 FEET.      C
C                                                                              C
CXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXC
C                                                                              C
C                                                                              C
C     --- DIMENSION ARRAYS ---                                                 C
C                                                                              C
      IMPLICIT REAL*8 (A-H,O-Z)
      INTEGER CVTOPS,BFTOPS,TREENO
      REAL*8 CVDIBS(10),CVHTS(10),CUVOLS(10)/10*0./,BFDIBS(10),
     1BFHTS(10),BFVOLS(10)/10*0./,TOPDIB(10),TOPHTS(10)
C                                                                              C
C                                                                              C
C     --- SPECIFY NUMBER OF MERCHANTABLE TOP DIBS TO BE USED                   C
C     CVTOPS = NUMBER OF CUBIC VOLUME MERCHANTABLE TOP DIBS                    C
C     BFTOPS = NUMBER OF BOARD FOOT MERCHANTABLE TOP DIBS                      C
C          IF CVTOPS = 0, THEN NO CUBIC VOLUMES WILL BE CALCULATED             C
C          IF BFTOPS = 0, THEN NO BOARD FOOT VOLUMES WILL BE CALCULATED ---    C
C                                                                              C
      CVTOPS=2
      BFTOPS=2
C                                                                              C
C     --- SPECIFY MERCHANTABLE TOP DIBS WITH DATA STATEMENT. LIST IN           C
C          DECENDING ORDER ---                                                 C
C                                                                              C
      DATA CVDIBS/4.,0./,BFDIBS/8.,6./
C                                                                              C
C     --- IN THIS EXAMPLE CUBIC VOLUMES INSIDE BARK TO A 4-INCH TOP,           C
C          AND TOTAL STEM WILL BE CALCULATED. BOARD FOOT VOLUMES TO            C
C          8- AND 6-INCH TOPS WILL BE CALCULATED ---                           C
C                                                                              C
C                                                                              C
CXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXC
C
C
C     --- READ TREE DBHOB AND TOTAL HEIGHT.
C          ALL READ AND WRITE STATEMENTS IN THIS PROGRAM ARE FOR SAMPLE
C          PROBLEM ONLY. IF NEEDED BY THE USER, THEY MUST BE MODIFIED ---
    5 READ(5,10,END=200)DBHOB,HT
   10 FORMAT(T15,F3.1,T21,F4.1)

C
C     --- CALCULATE DBHIB. CONSTANT DBHIB/DBHOB RATIO = .91 ASSUMED
C          FOR THIS EXAMPLE ---
      DBHIB=DBHOB*.91
C
C     --- CALL SUBROUTINES TO CALCULATE VOLUMES ---
      IF(CVTOPS.EQ.0)GO TO 20
      CALL BISECT(DBHIB,HT,CVTOPS,CVDIBS,CVHTS)
      CALL CUVOL(DBHIB,HT,CVTOPS,CVHTS,CUVOLS)
   20 IF(BFTOPS.EQ.0)GO TO 30
      CALL BISECT(DBHIB,HT,BFTOPS,BFDIBS,BFHTS)
      CALL BFVOL(DBHIB,HT,BFTOPS,BFDIBS,BFHTS,BFVOLS)
```

```
C
   30 WRITE(6,2)DBHOB,HT,(CUVOLS(I),I=1,CVTOPS),(BFVOLS(I),I=1,BFTOPS)
    2 FORMAT(T10,8F8.2)
      GO TO 5
  200 STOP
      END

      SUBROUTINE BISECT(DBHIB,HT,NTOPS,TOPDIB,TOPHTS)
C
C     --- THIS SUBROUTINE USES A BISECTION ALGORITHM (ITERATIVE
C         APPROACH) TO DETERMINE THE LENGTHS (TOPHTS) FROM TIP TO THE
C         DESIRED MERCHANTABLE INSIDE BARK DIAMETERS (TOPDIB) ---
C
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 TOPDIB(10),TOPHTS(10)
      A=0.0
      B=HT-1.0
      D=0.0
      ZMDIB=0.0
      DO 10 I=1,NTOPS
      K=I
      IF(TOPDIB(I).LT..01)GO TO 55
      Z=HT-1.0
      CALL DIB(DBHIB,HT,Z,D)
      IF(D.LE.TOPDIB(I))GO TO 40
      DO 20 J=1,50
      ZMIDPT=(A+B)/2.
      CALL DIB(DBHIB,HT,ZMIDPT,ZMDIB)
      DIFF=TOPDIB(I)-ZMDIB
      IF(DABS(DIFF).LT..01)GO TO 45
      IF(DIFF.LT.0.)GO TO 50
      A=ZMIDPT
      GO TO 20
   50 B=ZMIDPT
   20 CONTINUE
   40 TOPHTS(I)=HT-1.0
      GO TO 10
   45 TOPHTS(I)=ZMIDPT

      A=0.0
   10 CONTINUE
      GO TO 60
   55 TOPHTS(K)=0.0
   60 RETURN
      END

      SUBROUTINE DIB(DBHIB,HT,ZHT,ZDIB)
C
C     --- THIS SUBROUTINE CALCULATES THE DIB (ZDIB) AT A
C         GIVEN LENGTH FROM THE TIP (ZHT) ---
C
      IMPLICIT REAL*8 (A-H,O-Z)
      Z=ZHT/(HT-4.5)
      X1=Z**1.5
      X2=X1-Z**3
      X3=X1-Z**30
      X4=X2*HT
      X5=X3*DBHIB
      X6=X5*HT
      X7=X4*DBHIB
      ZDIB=X1+(3.04392D-3)*X4+(3.21012D-3)*X5-(2.12D-4)*X6+(2.2603D-4)*
     1X7
      ZDIB=DBHIB*(DSQRT(ZDIB))
      RETURN
      END
```

```
      SUBROUTINE CUVOL(DBHIB,HT,NTOPS,TOPHTS,CUVOLS)
C
C     --- THIS SUBROUTINE CALCULATES CUBIC VOLUMES (CUVOLS) TO
C         DESIRED MERCHANTABLE INSIDE BARK DIAMETERS.
C         TOPHTS = DISTANCE FROM TIP TO MERCHANTABLE HEIGHT ---
C
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 TOPHTS(10),CUVOLS(10)
      L=0
    5 DO 50 I=1,NTOPS
      K=I
      IF(TOPHTS(I).LT..001)GO TO 55
      IF(L.EQ.1)GO TO 10
      Z=TOPHTS(I)/(HT-4.5)
   10 X1=Z**2.5
      X2=Z**4
      X3=Z**31
      X4=X1*HT
      X5=X2*HT
      X6=X4*DBHIB
      X7=X5*DBHIB
      X8=X1*DBHIB
      X9=X3*DBHIB
      X10=X9*HT
      F=.4*X1
      F=F+(1.21757D-3)*X4
      F=F-(7.6098D-4)*X5
      F=F+(9.041D-5)*X6
      F=F-(5.651D-5)*X7
      F=F+(1.28405D-3)*X8
      F=F-(1.0355D-4)*X9
      F=F-(8.48D-5)*X6
      F=F+(6.84D-6)*X10
      IF(L.EQ.1)GO TO 60
      CUVOLS(I)=.00545415*DBHIB*DBHIB*(HT-4.5)*F
   50 CONTINUE
      GO TO 57
   55 CUVOLS(K)=0.0
   57 Z=(HT-1.0)/(HT-4.5)
      L=1
      GO TO 5
   60 TOTVOL=.00545415*DBHIB*DBHIB*(HT-4.5)*F
C
C     --- CALCULATE 1.0 FOOT STUMP VOLUME ---
      STUMPH=HT-1.0
      CALL DIB(DBHIB,HT,STUMPH,STUMPD)
      STUMPV=.00545415*STUMPD*STUMPD
C
C     --- CALCULATE CUBIC VOLUMES TO MERCHANTABLE TOP DIAMETERS ---
      DO 70 I=1,NTOPS
      CUVOLS(I)=TOTVOL-CUVOLS(I)
      IF(TOPHTS(I).LT..001)CUVOLS(I)=TOTVOL+STUMPV
   70 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE BFVOL(DBHIB,HT,NTOPS,BFDIBS,TOPHTS,BFVOLS)
C
C     --- THIS SUBROUTINE CALCULATES INT'L 1/4-INCH RULE BOARD
C         FOOT VOLUMES (BFVOLS) TO DESIRED MERCHANTABLE INSIDE BARK
C         DIAMETERS (BFDIBS). TOPHTS = DISTANCE FROM TIP TO MERCHANTABLE
C         HEIGHT ---
C
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 TOPHTS(10),BFVOLS(10),BFDIBS(10),LOGDIB
      LOGDIB=0.0
      DO 90 I=1,NTOPS
      TOPHTS(I)=HT-TOPHTS(I)
      H=1.0
      BFVOLS(I)=0.
   50 H=H+.3
      IF(H.GE.TOPHTS(I))GO TO 90
      H=H+16.
      IF(H.GT.TOPHTS(I))GO TO 55
      Z=HT-H
      CALL DIB(DBHIB,HT,Z,LOGDIB)
      BFVOLS(I)=BFVOLS(I)+.796*LOGDIB*LOGDIB-1.375*LOGDIB-1.23
      GO TO 50
   55 H=H-16.
      A=TOPHTS(I)-H
      ILOG4=A/4.
      LOGDIB=BFDIBS(I)
      IF(ILOG4.EQ.0)GO TO 65
      DO 60 J=1,ILOG4
      BFVOLS(I)=BFVOLS(I)+.199*LOGDIB*LOGDIB-.642*LOGDIB
      LOGDIB=LOGDIB+.5
   60 CONTINUE
   65 FRAC=(A-(FLOAT(ILOG4)*4.))/4.
      BFVOLS(I)=BFVOLS(I)+(.199*LOGDIB*LOGDIB-.642*LOGDIB)*FRAC
   90 CONTINUE
      RETURN
      END
```

VOLUMES GENERATED FOR FIVE SAMPLE TREES:

DBH_{OB}	TOTAL HEIGHT	CV4	CVTS	BF8	BF6
10.0	67	12.89	14.22	27.08	53.98
11.7	74	19.68	21.18	74.07	94.67
13.3	83	28.39	30.13	131.81	149.07
15.2	92	40.66	42.74	214.22	227.81
18.0	87	53.89	56.64	303.18	309.51

ACKNOWLEDGEMENTS

The author thanks Fred Berry, Northeastern Forest Experiment Station, and Will Carmean, North Central Forest Experiment Station, for use of their felled-tree measurements of Central States upland oaks. Thanks also to Jeff Martin, Northeastern Forest Experiment Station, for supplying the test data set from Virginia and West Virginia.

Headquarters of the Northeastern Forest Experiment Station are in Broomall, Pa. Field laboratories and research units are maintained at:

- Amherst, Massachusetts, in cooperation with the University of Massachusetts.
- Beltsville, Maryland.
- Berea, Kentucky, in cooperation with Berea College.
- Burlington, Vermont, in cooperation with the University of Vermont.
- Delaware, Ohio.
- Durham, New Hampshire, in cooperation with the University of New Hampshire.
- Hamden, Connecticut, in cooperation with Yale University.
- Kingston, Pennsylvania.
- Morgantown, West Virginia, in cooperation with West Virginia University, Morgantown.
- Orono, Maine, in cooperation with the University of Maine, Orono.
- Parsons, West Virginia.
- Princeton, West Virginia.
- Syracuse, New York, in cooperation with the State University of New York College of Environmental Sciences and Forestry at Syracuse University, Syracuse.
- University Park, Pennsylvania, in cooperation with the Pennsylvania State University.
- Warren, Pennsylvania.